
News 
Training Days  

Once again this year, we will be holding our US Spike2 and Signal training course on 
Thursday 12th and Friday 13th of October, prior to the Society for Neuroscience meeting in 
Atlanta. The venue will be the Sheraton Atlanta Hotel. These training days are for both 
existing and prospective users of Spike2 and Signal and cater for the beginner – with 
introductions to the software, through to advanced users – with sessions on script writing 
and output stimulus generation and control. There is our popular Workshop thread where 
you can discuss your particular requirements with a CED engineer or programmer in a 
small group environment. 

If you or any of your colleagues are interested in attending the course, please follow the 
links to view the training day program and complete the registration form. Please give a 
short description of the type of research that you do and analysis that you are interested 
in. This will allow us to tailor sessions for specific requirements. 

Meetings and events 
Society for Neuroscience Annual Meeting 2006 
Atlanta 
October 15th – 18th  

American Heart Association Scientific Sessions 2006 
Chicago 
November 12th – 14th 

 
  Latest versions of Spike2 and Signal 

The latest updates for Spike2 and Signal are available from the CED downloads page, or 
by clicking on the links in the table below. Demonstration versions of the latest software 
are also available. Spike2 version 6.01 will soon be ready for release as the first revision 
update for registered users. 
 

 Spike2 downloads Signal downloads 
Spike2 version 5.15  Signal version 3.07 

 Spike2 version 4.24  Signal version 2.16 

 Spike2 v6 demo  Signal demo 

A full list of the new features and changes in the latest software versions is available from 
the website. 

Spike2   

 

Q. I use active cursors to automate measurements from my data and would like a 
way to copy cursor settings to each file in a given folder. 

A. Spike2 data files are saved with an associated resource file. This will have the 
same name but with the extension .s2r. These resource files contain information 
about the channel display, cursor positions, active cursor parameters and other 
settings so that the file can be re-opened with exactly the same screen 
appearance and cursor settings it had when the file was closed. Using the Global 
resources command from the File menu it is possible to nominate a single 
resource file to apply to multiple data files. This is particularly useful when 
reviewing data stored on a read-only medium such as CD’s or DVD’s, as Spike2 
is not able to update or add a resource file. 

For this example, to apply settings from one file to all data files in a given folder, 
open the resources dialog and enable it by checking the ‘Use global…’ checkbox. 
You can then enter the name of the resource file you want to use (this will have 
the same name as the file you want to copy the settings from) and set the 
resource file location from the drop-down list. Additional fields in the dialog allow 
you to specify which files should have the resources applied. Set the data file 
path to only apply settings to files in a single folder. 

 
Global resources dialog 

   

 

  

Q. I am using a Micro1401 and Signal for EMG recording and generation of 
precisely timed TMS pulses. I have written a script which controls different 
recording conditions and I am now looking for a way to pause and continue the 
script via a foot switch while recording. Is there any way to do this?  

A. User-defined toolbars can be created via the script language to give control and 
allow interaction with the running script. Buttons created in the toolbar can be 
linked to functions in the script such as starting and stopping sampling and can 
be linked to hotkeys on the keyboard for easier access. The attached example, 
KeyboardPause.sgs, generates a toolbar with buttons to create a new file and 
pause and resume sampling. This pause button is linked to the Spacebar on the 
keyboard so the pause and resume can be controlled from a single keypress. 

 There are foot switches available that can generate keyboard characters via a 
USB connection to the host computer and these could be used to control button 
presses in a script. We currently have a customer testing a foot switch control; 
example equipment can be found here to give entry of up to 4 individual keys.      

'$Keyboard pause.sgs|Signal scritp to allow the user to stop and restart sampling from a keypress
'Uses a toolbar with a shortcut button (spacebar in this case) to stop sampling. Button is then reset to 
	'run a different function to continue the sampling
'Uses the current sampling configuration


var data%;

View(App(3));			'make script window the current view
WindowVisible(0);		'hide script window

DoToolbar();   'Try it out
Halt;

Func DoToolbar()    'Set your own name...
ToolbarClear();     'Remove any old buttons
ToolbarSet(1, "Quit");    'This button returns its number
ToolbarSet(3, "New file", file%);   'Link to function
ToolbarSet(5, "Start sampling", sample%);   'Link to function
ToolbarSet(7, "Pause|0x20",Pause%);	'Link to function and set spacebar as shortcut
ToolbarEnable(5,0);					'Tidy up toolbar	
ToolbarEnable(7,0);
return Toolbar("Your prompt",1023);		'Allow certain activities
end;



Func file%()   'Button 3 routine
data%:=Filenew(0);					'New file
Window(0,0,100,100);
WindowVisible(1);					'Full page
View(data%);
If data%<0 then						'Did file open?
	Message("No file opened");
Halt;
Endif;
ToolbarEnable(3,0);					'Tidy up toolbar
ToolbarEnable(5,1);
ToolbarEnable(7,1);
return 1;  'This leaves toolbar active
end;

Func sample%()   'Button 5 routine
SampleStart();						'Start sampling
ToolbarSet(5, "End sampling",Stop%);
ToolbarEnable(7,1);					'Tidy up toolbar
return 1;  'This leaves toolbar active
end;



Func stop%()	'Button 5 routine
SampleStop();						'Stop sampling
ToolbarSet(5, "Start sampling", sample%); 
ToolbarEnable(3,1);	
ToolbarEnable(5,0);					
ToolbarEnable(7,0);
return 1;
end;


Func Pause%()	'Button 8 routine
View(data%);
SampleStop(1);										'Stop samling (Could be replaced with SamplePause(1))
ToolbarSet(7, "Resume|0x20",Resume%);				'reset button to link to Resume%() function
return 1;
end;

Func Resume%()	'Button 8 routine
View(data%);
SampleSweep();										'Continue sampling (If SamplePause approach used this should be changed to SamplePause(0)	
ToolbarSet(7, "Pause|0x20",Pause%);
return 1;
end;

Scripts: Spike2  
Q. Is it possible to apply a time shift function to event and marker channels? I know 

how to do this to waveform channels using the Channel Process functions but 
would like to be able to shift events as well.   

A. The Channel Process functions can only be applied to waveform channels, as 
often they directly affect the data point values. The attached script, 
TimeShiftEvents.s2s can be used to copy event or marker times from the 
currently selected data file to a new permanent channel in any open data file. 
The script creates 2 cursors with which you select the data range to copy and 
you can apply an optional time shift by selecting the start time for the data in the 
new channel. 

 

Scripts: Signal  

Q. I would like to be able to produce spreadsheet text output of a nominated 
channel that includes all overdrawn frames. 

A. The attached script, OverdrawnAsText.sgs will create a spreadsheet format 
text file of data from all frames in the frame display list of the current data file. 
The user nominates the channel to extract the text values from and the script 
collects the channel data from all frames in the display list. The user is then 
prompted to save the spreadsheet output as a text file, which can then be 
opened in Signal or in a spreadsheet program. 

Did you know…? 
You can create or fetch vertical cursors in Spike2 by holding Ctrl and pressing the 
appropriate number key on the keyboard. For example, Ctrl + 1 will create or fetch 
Cursor(1) in the current view. In Signal vertical cursors are created using the Ctrl + | 
key in compatible view types. 

Recent questions  
Q. I have just recently joined a research group that has a Power1401. There are 

only 8 ADC inputs on the front of the unit and I would like to use up to ten 
channels. Is there any way to expand this unit to include more channels?  

A. The Power1401 comes with 16 ADC channels as standard. 8 of these are found 
on the front of the unit as BNC connectors and a further 8 channels are available 
as pin inputs on the rear analogue expansion port. This rear expansion port also 
includes two additional DAC outputs to give 4 in total. We offer several expansion 
options in the form of top-boxes for users who require more inputs and outputs 
than are available on the Power1401 main unit. Top-boxes that offer ADC and 
DAC channels are always in addition to those already available on the standard 
unit, so, 8 ADC inputs and 2 DAC outputs are always available for use from the 
analogue expansion port. 

      

   Power1401 625kHz analogue expansion socket 

scripts in this newsletter 
please let us know.  

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top

Back to top



'$Time shift events|Save channel of events or markers with optional offset


'	===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB0 0FE, UK
'	===========================================================================================================


'	Copyright © Cambridge Electronic Design, August 2006.
'	Author:	GH.

'	SOFTWARE REQUIRED: Spike 2 version 5.15 or higher

'	HEALTH WARNING:
'	The script is a 'work in progress' and is offered without guarantees.
'	Make backup copies of your results before using the script and check
'	carefully that the script generates the results that you expect.

'   OVERVIEW
'   Script to copy an Event or Marker channel to a new permanent channel in the original data file
'   or any other open time view. You can apply a time shift to the data by selecting the start time of the
'   copied data. 
'   Run this script using the Load and Run option on the Spike2 Script drop-down menu or via a button on the Script bar.

'   USER GUIDE:
'   Click on the time view containing the Event/Marker to be shifted in order to make it the front view before running the script.
'   A dialog will open for you to select the source channel, the time range to copy, the destination for the results
'   and the new start time for data. You can select cursors to bracket the time range if you wish.
'   The script will then create the new channel in the destination file or report any problems that were encountered.
'   If you wish to copy a channel from one data file to another without any time shift, simply set the
'   same start time for source and destination.



var tvh%,ok%,i%,bfr,ch%;
var stime,etime,dtime,srcch%,destvh%,destndx%;
var vlist%[20],vlist$[20],lst%[2];

View(App(3)).WindowVisible(2); 'iconise script
tvh%:=FrontView(); 'check that the front view is a time view
docase
case ViewKind(tvh%) <> 0 then
    Message("Front view is not a time view. Click on the source time view and try again.");
    halt
case ChanList(lst%[],2+4+8+16+32+2048+16384) = 0 then 
    Message("No visible marker or event channel found in the source file: <%s>.",WindowTitle$());
    halt
endcase;

ViewList(vlist%[],1);'list of open time views
for i%:=1 to vlist%[0] do 'create list of time view window titles for dialog
    View(vlist%[i%]);
    vlist$[i%-1]:=WindowTitle$();
    if vlist%[i%] = tvh% then
        destndx%:=i%-1;'set default destination view to same as source view
    endif;        
next;

View(tvh%);
if CursorExists(1) = 0 or CursorExists(2) = 0 then 'create 2 cursors unless we already have some
    CursorSet(2); 'option to set time range with cursors
else
    CursorRenumber();
endif;
stime:=Cursor(1);
etime:=Cursor(2); 'default time range

DlgCreate("Time shift Markers");
DlgAllow(511);'allow cursor dragging (and much more) while the dialog is showing
DlgGroup("Source:",1,1,38,5);
DlgText(Print$("View:                %s",WindowTitle$()),2,2);
DlgChan(1,"Channel",2+4+8+16+32+2048+16384,15,3); 'visible markers and events (not duplicates)
DlgXValue(2,"Start time:",20,4);
DlgXValue(3,"End time:",20,5);
DlgGroup("Destination:",1,6.5,38,3);
DlgText("View:",2,7.5);
DlgList(4,28,vlist$[:vlist%[0]],vlist%[0],10,7.5);
DlgReal(5,"Start time (s):",0.0,10e12,26,8.5);
ok%:=DlgShow(srcch%,stime,etime,destndx%,dtime);
if not ok% then halt endif; 'cancelled

'error checking
if stime > etime then
    bfr:=etime; etime:=stime; stime:=bfr; 'flip start and end times
    halt
endif;
if Count(srcch%,stime,etime) = 0 then 'is there anything to copy?
    Message("No events /Markers in the selected time range.");
    halt
endif;

destvh%:=vlist%[destndx%+1]; 'get view handle of chosen destination
View(destvh%);
docase
case dtime > MaxTime() then
        ok%:=Query("Start time is after the current MaxTime in the destination file.","Continue","Cancel");
    if not ok% then
        halt
    endif;
case dtime+etime-stime > Maxtime() then 'Is there room for all the events if we start at the requested time?
    ok%:=Query("Time shift extends beyond MaxTime in the destination file.","Continue","Cancel");
    if not ok% then
        halt
    endif;
endcase;

View(tvh%);
ch%:=ChanSave(srcch%,0,destvh%,stime,etime,dtime); 'copy channel with time shift
if ch% < 0 then
    Message(Error$(ch%));
else
    View(destvh%);
    ChanShow(ch%); 'show it
endif;
halt


david
timeshiftEvents.s2s


'$OverdrawnAsText|Script to give a spreadsheet text style output of overdrawn frames, with one
'frame per column. User selects a waveform channel from an open data file and waveform values
'for each frame are collected and saved to an external text file named by the user.

var file%;					'Handle for file 
var nFrame%;				'Number of frames in file
var frames%[500];			'Array to hold frames from overdraw list

if ViewKind() = 0 then		'Check selected view holds data
	
file%:= View();				'Get the file handle
Window(0,0,100,100);		'Position window
Frame(1);					'Go to first frame	
FrontView(file%);			'Bring to front
Optimise(-1);				'Optimse the data
nFrame%:=OverdrawGetFrames(frames%[]);		'Count frames in overdraw list
if nFrame% = 0 then			'If nop frames in list
Message("No frames are selected in the overdraw frame list! Set frames for overdraw and re-run the script");	'warn user and halt
halt;endif;	
else
Message("No data file selected! Select a data file and re-run script!");halt;		'If current view is not a data file, warn user and halt
endif;

PrintVals%();		'Call the function for text output

Func PrintVals%()
var chan%;			'Waveform channel to print
var nPoints%;		'No. of data points in waveform channel
var c%;				'Counter for frames
var n%;				'Counter for data points
var txt%;

DlgCreate("Select channel to print");	'Prompt user to select a channel
DlgChan(1,"Channel",1);
DlgShow(chan%);
	
nPoints%:= (1/BinSize(chan%))*(MaxTime()-MinTime());	'Get the number of data points

var frameVals[nPoints%+1][nFrame%+1];	'Create an array to hold data points and frame number

for c%:= 1 to nFrame% do		'For each overdrawn frame
	Frame(frames%[c%]);			'Set frame to current

	for n%:= 1 to nPoints% do	'For each data point
		frameVals[n%][c%]:= View(file%,chan%).[n%-1];	'get the values for the given frame
	next;

next; 	

txt%:= FileOpen("",8,1);	'Open external text file

FrontView(txt%);

Print("s");		'Column for time of point

for c%:= 1 to nFrame% do		'For each frame
	Print("\tFrame %d",frames%[c%]);		'Print column headers
next;
	
for n%:=1 to nPoints% do		'For each data point
	Print("\n%f",(View(file%).BinSize(chan%)*(n%-1)));	'Print time of current point

	for c%:= 1 to nFrame% do	'For each frame	
		Print("\t%f",frameVals[n%][c%]);		'Print data values
	next;

next;

return 1;
end;

User Group  

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

